

# Alternative Methods for Implementing Explicit and Finding Implicit Sharing in embedded DSLs

Curtis D’Alves, Christopher Anand, Lucas Dutton, and Steven Gonder

McMaster University, 1280 Main St W Hamilton, Canada

**Abstract.** TODO The abstract should briefly summarize the contents of the paper in 150–250 words.

**Keywords:** First keyword · Second keyword · Another keyword.

## 1 Introduction

TODO describe sharing problem (mention observable sharing [4] and implicit/-explicit sharing [5] papers)

TODO describe finally tagless [1]

We present methods for implementing embedded DSLs with sharing that are both safe and maintain all the benefits of being embedded in the Haskell ecosystem. This means DSL functions are type-safe, do not require the use of unsafe referencing (i.e., via `unsafePerformIO`) and can return Haskell’s container types (i.e., tuples, lists, etc) without breaking sharing.

## 2 Detecting Sharing

A naive DSL implementation of an expression in Haskell can be done via standard Haskell data types, for example:

```
data Exp
  = Add Exp Exp
  | Variable String
  | Constant Int

-- Example
v0 = Variable ``v0''
exp0 = Add v0 (Constant 0)
exp1 = Add exp0 exp0
```

Note the DSL generates a tree, or to be more specific an Abstract Syntax Tree (AST). Common features a DSL implementer might implement would include code generation or pretty printing. Simple traversal of the AST for either of these operations would result in duplication, for example in the above code snippet the AST for **exp0** will be traversed twice. For code generation in particular this

would be problematic, in order to circumvent this problem in general we need to perform common subexpression elimination by converting the AST into a directed acyclic graph.

TODO describe DAG conversion by pointer comparison in the state monad

## 2.1 Detecting Sharing In Finally Tagless DSLs

Monads are useful, but don't make for a very user friendly DSL. It would be nice to make use of monadic state when we need it (i.e., for converting to a DAG) while hiding it behind a nice pure interface. The final tagless approach of [1] is popular for accomplishing this. In this approach, DSL expressions are built using typeclass methods that wrap the DSL in a parameterized representation. For example, the previous data type based DSL could be written in finally tagless as

```
class Exp repr where
  add :: repr Int -> repr Int -> repr Int
  variable :: String -> repr Int
  constant :: Int -> repr Int
```

We can then create different instances to implement different functionality. For example, we can generate the AST from the previous DSL like so

```
newtype Expr a = Expr { unExpr :: Exp }
```

```
instance Exp Expr where
  constant = Expr . Constant
  variable = Expr . Variable
  add (Expr x) (Expr y) = Expr (Add x y)
```

Or we can implement pretty printing

```
newtype Pretty a = Pretty { unPretty :: String }
```

```
instance Exp Pretty where
  add x y = Pretty $ "("++unPretty x++") + ("++unPretty y++")"
  variable x = Pretty x
  constant x = Pretty $ show x
```

And use the same DSL code to run either implementation

```
exp :: Exp repr => repr Int -> repr Int
exp v0 =
  let
    exp0 = add v0 (constant 0)
  in add exp0 exp0

expr = unExpr $ exp $ variable "v0"
expP = unPretty $ exp $ variable "v0"
```

Finally tagless style provides extensible, user friendly DSLs. However there are still some complications when using it to implement sharing.

## 2.2 Implicit Sharing Via Hash-Consing

TODO cite Ershov's original description of hash-consing [2] cite Type safe consing implementation (with performance benchmarks) [3]

In [5], a solution for detection implicit sharing in finally tagless style is presented via the method of hash-consing. You can find a more thorough explanation of the method there, but we'll give an overview here. This method first involves defining a DAG type, for example

```
type NodeID = Int
data Node = NAdd NodeID NodeID
          | NVariable String
          | NConstant Int

data BiMap a -- abstract
lookup_key :: Ord a => a -> BiMap a -> Maybe Int
lookup_val :: Int -> BiMap a -> a
insert :: Ord a => a -> BiMap a -> (Int, BiMap a)
empty :: BiMap a

newtype DAG = DAG (BiMap Node) deriving Show
```

Note the purpose of the BiMap type is to be able to quickly insert and lookup nodes by their NodeID (i.e., a bijection of Node's and their NodeID's), and is most optimally implemented as a hash table with linear probing. The representation for the finally tagless instance is then a wrapper around a State monad that holds DAG in its state and returns the current (top) NodeID.

```
newtype Graph a = Graph { unGraph :: State DAG NodeID }

instance Exp Graph where
  constant x = Graph (hashcons $ NConstant x)
  variable x = Graph (hashcons $ NVariable x)
  add e1 e2 = Graph (do
    h1 <- unGraph e1
    h2 <- unGraph e2
    hashcons $ NAdd h1 h2)
```

The trick to uncovering sharing in the implementation is implemented via the **hashcons** function, which inserts a new node into the current DAG, but not before checking if it is already there.

```
hashcons :: Node -> State DAG NodeID
hashcons e = do
```

```

DAG m <- get
case lookup_key e m of
  Nothing -> let (k,m') = insert e m
              in put (DAG m') >> return k
  Just k -> return k

```

### 2.3 Limitations of Hash-Consing

When we wrap our State monad in finally tagless style, we lose some of Haskell’s built-in sharing capability. Consider the following code, note that the use of local variables explicitly defines the computation  $x + y$  to only be computed once

```

haskellSharing x y =
  let
    z = x + y
  in z + z

```

Implicit sharing via hash-consing prevents duplication in the resulting DAG, but unfortunately doesn’t prevent redundant computation. Consider the following equivalent attempt at using Haskell’s built-in sharing in the finally tagless DSL

```

dslSharing :: Exp Graph -> Exp Graph -> Exp Graph
dslSharing x y =
  let
    z = add x y
  in add z z

```

Note **z** is a wrapper around a State monad. Recall the implementation of **add** via hash consing

```

add e1 e2 = Graph (do
  h1 <- unGraph e1
  h2 <- unGraph e2
  hashcons $ NAdd h1 h2)

```


The values **h1** and **h2** need to be explicitly evaluated through the State monad, meaning even if **e1** and **e2** are the same shared Haskell value, their underlying computations will be performed twice. Hash-consing will prevent these redundancies from appearing in the resulting DAG, but the entire unshared AST will still be traversed, performing a hash-cons on each node.

Consider a chain of **add**’s with sharing, for example

```

addChains :: Exp repr => Expr Int -> Expr Int
addChains x0 = let
  x1 = add x0 x0
  x2 = add x1 x1
  ...
in xn

```



**Fig. 1.** Number of hashcons to add operations performed

As you can see from figure 1, this code will perform approximately  $2^{n+1}$  hashcons operations, where  $n$  is the number of `add`'s.

## 2.4 Explicit Sharing and Limitations

[5] acknowledges the amount of computation with hash-consing can get out of control, and proposes an ad-hoc solution, explicit sharing via a custom `let` construct

```
class ExpLet repr where
  let_ :: repr a -> (repr a -> repr b) -> repr b
instance ExpLet Graph where
  let_ e f = Graph (do x <- unGraph e
                      unGraph $ f (Graph (return x)))

addChains x =
  let_ x (\x0 ->
  let_ (add x0 x0) (\x1 ->
  let_ (add x1 x1) (\x2 ->
    ...
  )))
```

This makes the code a bit more clunky and adds an extra burden on the DSL writer, but it prevents unnecessary hash-consing in our example. However the method has its limitations, suppose we want to extend our DSL to support functions that return multiple outputs, such as tuples or container types like lists (i.e. `vadd :: (repr Int, repr Int) -> (repr Int, repr Int)`). First of all, we need to implement different versions of the custom `let` construct to correspond to the number of outputs

```

class ExpLet repr where
  let_ :: repr a -> (repr a -> repr b) -> repr b
  let_ 2 :: repr a -> (repr a -> (repr b,repr c)) -> (repr b,repr c)

instance ExpLet repr where
  ...
  let_2 e f = let
    g0 = Graph (do x <- unGraph e
                  let (o0,o1) = f (Graph (return x))
                  unGraph o0)
    g1 = Graph (do x <- unGraph e
                  let (o0,o1) = f (Graph (return x))
                  unGraph o1)
    in (g0,g1)

```

If need be, its possible to enumerate custom let instances for every amount of outputs or container types we would need, we could even use template Haskell to accomplish this. However, this custom let construct now has a new source of redundancy in its outputs. Each output it returns will now have to individually evaluate it's input, so a chain of DSL functions that output 2 or more values will suffer from the same exponential scaling of hashcons.

### 3 Implicit Sharing Via ByteString ASTs

TODO

#### 3.1 Memory Limitations

TODO

### 4 Explicit Sharing Of ByteString ASTs

TODO

**Acknowledgements** Please place your acknowledgments at the end of the paper, preceded by an unnumbered run-in heading (i.e. 3rd-level heading).

### References

1. Carette, J., Kiselyov, O., Shan, C.c.: Finally tagless, partially evaluated. In: Asian Symposium on Programming Languages and Systems. pp. 222–238. Springer (2007)
2. Ershov, A.P.: On programming of arithmetic operations. Communications of the ACM **1**(8), 3–6 (1958)

3. Filliâtre, J.C., Conchon, S.: Type-safe modular hash-consing. In: Proceedings of the 2006 Workshop on ML. pp. 12–19 (2006)
4. Gill, A.: Type-safe observable sharing in haskell. In: Proceedings of the 2nd ACM SIGPLAN symposium on Haskell. pp. 117–128 (2009)
5. Kiselyov, O.: Implementing explicit and finding implicit sharing in embedded dsls. arXiv preprint arXiv:1109.0784 (2011)